

Dioids for computational effects

Ezequiel Postan
DCC–ECEN

Facultad de Cs. Exactas, Ingeniería y Agrimensura
Universidad Nacional de Rosario

Exequiel Rivas and Mauro Jaskelioff
DCC–ECEN

Facultad de Cs. Exactas, Ingeniería y Agrimensura
Universidad Nacional de Rosario
and
CIFASIS–CONICET

Abstract—There are different algebraic structures that one can use to model notions of computation. The most well-known are monads, but lately, applicative functors have been gaining popularity. These two structures can be understood as instances of the unifying notion of monoid in a monoidal category. When dealing with non-determinism, it is usual to extend monads and applicative functors with additional structure. However, depending on the desired non-determinism, there are different options of interaction between the existing and the additional structure. This article studies one of those options, which is captured algebraically by dioids. We generalise dioids to dioid categories and show how dioids in such a category model non-determinism in monads and applicative functors. Moreover, we study the construction of free dioids in a programming context.

I. INTRODUCTION

Algebraic structures have been central to the modelling of computational effects. For example, monads [1], [2], [3], [4] have been used to model many computational effects such as global state, exceptions, environments, input/output, and continuations. More recently, applicative functors [5] are becoming popular in diverse applications such as modelling parsers [6], characterising traversals [7], [8], and in combination with monads to obtain concurrent queries [9].

While monads and applicative functors are two different algebraic structures, they have a unified framework. Rivas and Jaskelioff [10] have shown that both of them can be seen as instances of a same unifying concept: monoids in a monoidal category [11]. This unification of concepts is extremely useful, as it allows us to translate concepts, properties, and techniques from one structure to the other. For example, through the unified framework, an old optimisation technique for lists [12] is shown to be essentially the same as a newer one for monads [13], and led to the discovery of a new one for applicative functors [10] by means of a simple translation.

In many applications of monads and applicative functors one has to deal with non-determinism. There are different flavours of non-determinism [14], but in functional programming the most common are deep backtracking and shallow backtracking [15]. When modelling deep backtracking, the algebraic structure that arises is near-semirings. This insight lead to a unified framework for deep-backtracking non-determinism in monads and applicative functors [16]. If, on the other hand, one wants to model shallow backtracking, then one arrives at the algebraic structure of dioids [17].

This article studies the shallow-backtracking variant of non-determinism by studying the categories that support the definition of dioids, namely dioid categories. Working at this level of abstraction allows us to obtain a unified model of shallow-backtracking non-determinism for both monads and applicative functors. Moreover, we study the construction of free dioids. Intuitively, free dioids can be thought of as the programs that can be written when only the dioid interface is exposed, and therefore provide a canonical representation for programs structured as a computation with shallow-backtracking non-determinism.

The article is structured as follows: In Section II we introduce monoids and monoidal categories, and show how they provide a unified framework to study the notions of monads and applicative functors. In Section III, we introduce dioids and dioid categories. Moreover, we show how these categories provide a unified framework to study shallow non-determinism in monads and applicative functors. In Section IV, we turn to the construction of free dioids. We provide a formula that allows to construct dioids on Set (the category of sets and functions) and to construct the free dioid applicative. Unfortunately, it does not allow us to express the free dioid monad. Finally, in Section V, we conclude.

In the rest of this article, unless we explicitly say otherwise, when we write non-determinism we mean shallow-backtracking non-determinism.

II. MONOIDS

We start by studying monoids and its generalisation: monoids in monoidal categories. In order to keep the ideas close to programming practice, we express the different concrete constructions in an idealised functional programming language with syntax inspired by Haskell.

A. Monoids in Programming

In functional languages, such as Haskell, algebraic structures may be implemented using type classes. For example, for monoids we may declare:

```
class Monoid m where
  u :: () → m
  (⊗) :: m × m → m
```

which means that a type m is a monoid whenever it is equipped with a chosen binary operation \otimes (the *multiplication*) and a nullary operation u (the *unit*). Instances of this class are expected to satisfy the monoid laws

$$a \otimes (u ()) = a \quad (1)$$

$$(u ()) \otimes a = a \quad (2)$$

$$a \otimes (b \otimes c) = (a \otimes b) \otimes c \quad (3)$$

which state that \otimes is associative, and that u is a right and left unit for it. For example, we may declare that the type Integer of arbitrary-precision integers is a monoid of addition and zero:

instance Monoid Integer where

`u () = 0`

`x \otimes y = x + y`

Another important example is the monoid of endofunctions over a type a . Without mathematical jargon, they are just functions from a to a with composition as multiplication and the identity function as unit.

data Endo a where

`Endo :: (a -> a) -> Endo a`

instance Monoid (Endo a) where

`u () = Endo id`

`(Endo f) \otimes (Endo g) = Endo (f \circ g)`

When studying an algebraic structure, an important concept is that of homomorphism: structure-preserving maps between instances of the algebraic structure. In the case of monoids, they are defined as follows: let M_1 and M_2 be instances of Monoid. A *monoid homomorphism* is a function $f :: M_1 \rightarrow M_2$ such that the monoid instances are preserved:

$$f (u ()) = u ()$$

$$f (a \otimes b) = f a \otimes f b$$

The type of f determines that the operations u and \otimes on the left-hand side come from the Monoid instance of M_1 , while those on the right-hand side come from the Monoid instance of M_2 .

For example, the following is a monoid homomorphism from Integer to Endo Integer.

`rep :: Integer -> Endo Integer`

`rep x = Endo (\lambda y -> x + y)`

We want to program with algebraic structures such as Monoid above, but we may also want to generalise from types to types constructors, and from the Cartesian product \times to other ways of putting two things together. The appropriate generalisation is the notion of monoid in a monoidal category.

B. Categorification

Monoidal categories generalise the notion of monoids to categories \mathcal{C} .

Definition II.1. A *monoidal category* is a sextuple $(\mathcal{C}, \otimes, I, \alpha, \lambda, \rho)$, where:

- \mathcal{C} is a category;
- $\otimes : \mathcal{C} \times \mathcal{C} \rightarrow \mathcal{C}$ is a bifunctor;
- I is an object of \mathcal{C} ;
The bifunctor \otimes and I generalise \times and $()$ in the code above.
- α , λ , and ρ are natural isomorphisms:

$$\alpha : A \otimes (B \otimes C) \cong (A \otimes B) \otimes C$$

$$\lambda : I \otimes A \cong A$$

$$\rho : A \otimes I \cong A$$

All these natural transformations should obey coherence laws [11].

Given a monoidal category, we can define the notion of a monoid in it.

Definition II.2. A *monoid in a monoidal category* is an object M , together with operations $e : I \rightarrow M$ and $m : M \otimes M \rightarrow M$, for which the following laws hold:

$$m \circ (e \otimes \text{id}) = \lambda \quad (4)$$

$$m \circ (\text{id} \otimes e) = \rho \quad (5)$$

$$m \circ (\text{id} \otimes m) = m \circ (m \otimes \text{id}) \circ \alpha \quad (6)$$

The laws for monoids in monoidal categories are the corresponding generalisations of equations (1), (2), and (3).

For example, the category Set of sets and functions is a monoidal category with the Cartesian product as its tensor, and singleton sets as unit object, and monoids in this monoidal category reduce to ordinary monoids.

C. Functors

In this article we are mainly interested in the category of endofunctors and natural transformations. Endofunctors are type constructors that can map a function on the underlying type. For example, lists are functors since

- 1) given a type, say Integer, they construct a type of list of Integers;
- 2) given a function, say from Integers to Booleans, they can apply the function to every Integer to obtain a list of Booleans.

More precisely, a functor is an instance of the following class

class Functor f where

`fmap :: (a -> b) -> f a -> f b`

where the laws `fmap id = id`, and `fmap (f \circ g) = fmap f \circ fmap g` hold.

For example, the identity type constructor is a functor.

data Identity where

`Id :: a -> Identity a`

instance Functor Identity where

`fmap f (Id a) = Id (f a)`

Furthermore, the composition of two functors is a functor.

data $(f \circ g) a$ **where**

Comp :: $f (g a) \rightarrow (f \circ g) a$

instance (Functor f , Functor g) \Rightarrow Functor $(f \circ g)$ **where**

fmap f (Comp x) = Comp (fmap (fmap f) x)

In this instance, to the left of the \Rightarrow symbol, we note the requirement that f and g must be functors.

Natural transformations are functions between functors which are polymorphic on the underlying type [18].

type $(f \dot{\rightarrow} g) = \forall x. f x \rightarrow g x$

As a last functor example, we consider the Maybe data type constructor. This functor is commonly used to represent either a value or a failure, and it is sometimes known as Option.

data Maybe a **where**

Nothing :: Maybe a

Just :: $a \rightarrow$ Maybe a

The functor instance simply maps a function if we have a value, or it does nothing otherwise.

instance Functor Maybe **where**

fmap f Nothing = Nothing

fmap f (Just x) = Just ($f x$)

D. Monads

The category of endofunctors and natural transformations can be given a monoidal structure by choosing the tensor \otimes to be the composition of functors \circ , and the object I to be the identity functor Identity. This monoidal category is *strict*, which means that the three natural transformations λ , ρ , and α , which complete the monoidal category, are identities.

The monoids in this monoidal category are functors m with operations e and m of type Identity $\dot{\rightarrow} m$, and $m \circ m \dot{\rightarrow} m$ respectively. Expanding the definitions of natural transformation, identity functor, and functor composition, and renaming e to η and m to μ , we arrive at the following type class:

class Functor $m \Rightarrow$ Triple m **where**

$\eta :: a \rightarrow m a$

$\mu :: m (m a) \rightarrow m a$

and the general monoid laws (4), (5), and (6) become:

$$\mu \circ \eta = \text{id}$$

$$\mu \circ \text{fmap } \eta = \text{id}$$

$$\mu \circ \text{fmap } \mu = \mu \circ \mu$$

A monoid in the monoidal category with functor composition as tensor is none other than a *monad*. Monads have other presentations. For example, in Haskell, monads are defined as:

class Monad m **where**

return :: $a \rightarrow m a$

(\succ) :: $m a \rightarrow (a \rightarrow m b) \rightarrow m b$

subject to some laws which correspond to the laws above under the following equivalence. The classes Triple and Monad can be seen to be equivalent by noting that $\eta = \text{return}$, that μ can be defined as $\mu x = (x \succ \text{id})$, and that (\succ) can be defined as $(x \succ k) = \mu (\text{fmap } k x)$. Notice that Monad m does not require the type constructor m to be a functor: the fmap operation is derivable from the Monad instance by defining $\text{fmap } f v = (v \succ \text{return } \circ f)$.

Remark II.3 (Currying). The function (\succ) in the Monad class does not take two arguments; instead it takes only one argument and returns a function which takes the other argument and finally delivers the result. Writing functions in this style is equivalent to writing functions with two arguments, as the following conversion functions show:

curry :: $(a \times b \rightarrow c) \rightarrow (a \rightarrow b \rightarrow c)$

curry $f a b = f (a, b)$

uncurry :: $(a \rightarrow b \rightarrow c) \rightarrow (a \times b \rightarrow c)$

uncurry $f (a, b) = f a b$

In the rest of the presentation we use two argument functions or their curried form indistinctly.

Both lists and Maybe are functors which are also Monad instances. We provide the instance for Maybe, which will be one of our main examples.

instance Monad Maybe **where**

return $x =$ Just x

Nothing $\succ f =$ Nothing

(Just x) $\succ f = f x$

The category of endofunctors can be given other monoidal structures, and therefore monads are not the only monoids in the category of endofunctors. Another important class are *applicative functors*, introduced by McBride and Paterson [5] as a way to capture certain effectful computations that do not fit well in the monadic framework.

E. Applicative functors

Applicative functors are based on a category of endofunctors, but with different tensor than monads: the *Day convolution* [19]. The Day convolution may be implemented as follows:

data $(\star) f g a$ **where**

Day :: $f (b \rightarrow a) \times (g b) \rightarrow (f \star g) a$

instance (Functor f , Functor g) \Rightarrow Functor $(f \star g)$ **where**

fmap h (Day $ff gx$) = Day (fmap ($\lambda f \rightarrow h \circ f$) ff) gx

Just like for composition of functors (\circ), the I object for Day convolution is the identity functor. However, in this case the monoidal category is not strict. The isomorphisms λ , ρ and α for this monoidal category are as follows (we give only one direction of each isomorphism).

$\lambda ::$ Functor $f \Rightarrow$ (Identity $\star f$) $a \rightarrow f a$

λ (Day (Identity f) x) = fmap $f x$

```

ρ :: Functor f => (f * Identity) a -> f a
ρ (Day f (Identity b)) = fmap (λh -> h b) f
α :: (Functor f, Functor g) =>
  (f * (g * h)) a -> ((f * g) * h) a
α (Day f (Day g z)) = Day (Day (fmap (◦) f) g) z

```

The monoids in this monoidal category are functors f with operations e and m of type $\text{Identity} \dot{\rightarrow} f$, and $f * f \dot{\rightarrow} f$ respectively. Expanding the definitions of natural transformation, identity functor, and Day convolution, and renaming e to pure and m to \otimes , we arrive at the following type class:

```

class Functor f => Applicative f where
  pure :: a -> f a
  (⊗) :: f (b -> a) × f b -> f a

```

Instantiating the general monoid laws (4), (5), and (6) to this monoidal category, we obtain the applicative laws:

```

pure f ⊗ u = fmap f u
u ⊗ pure x = fmap (λh -> h x) u
(fmap (◦) u ⊗ v) ⊗ w = u ⊗ (v ⊗ w)

```

By generalising monoids to monoids in monoidal categories, we were able to show that two different structures used in programming are instances of the same abstract construction.

Remark II.4. Monad and Applicative type classes are not totally independent: every instance of the former is an instance of the latter, as it is reflected in the next code.

```

instance Monad m => Applicative m where
  pure x = return x
  u ⊗ v = u >= λf -> v >= λx -> return (f x)

```

Therefore every Monad determines an Applicative, but not the other way. As an example of an Applicative which is not a Monad, consider the constant functor on a monoid:

```

data K x a where
  MK :: x -> K x a
instance Functor (K x) where
  fmap f (MK x) = MK x
instance Monad x => Applicative (K x) where
  pure a = MK u
  (MK x) ⊗ (MK y) = MK (x ⊗ y)

```

III. DIOIDS

We extend the notion of monoids in order to account for non-determinism. More precisely, we introduce *dioids*, which extend monoids with additional monoid operations, which we denote with \oplus and z . Whereas the multiplicative monoid gives a model of sequencing, the \oplus operation models a non-deterministic choice, and models z the absence of choice. What makes this structure a good model for shallow-backtracking non-determinism, and what differentiates it from other models, is its interaction with the existing monoid structure.

A. Set dioids

Dioids are an algebraic structure, so we might declare them as a type class, just like we did with monoids:

```

class Dioid d where
  z :: () -> d
  u :: () -> d
  (⊕) :: d × d -> d
  (⊗) :: d × d -> d

```

This time, we expect the following laws to be satisfied:

$$a \otimes (u ()) = a \quad (7)$$

$$(u ()) \otimes a = a \quad (8)$$

$$a \otimes (b \otimes c) = (a \otimes b) \otimes c \quad (9)$$

$$a \oplus (z ()) = a \quad (10)$$

$$(z ()) \oplus a = a \quad (11)$$

$$a \oplus (b \oplus c) = (a \oplus b) \oplus c \quad (12)$$

$$(z ()) \otimes a = z () \quad (13)$$

$$(u ()) \oplus a = u () \quad (14)$$

The laws 7 to 12 express that d is a monoid with respect to (\otimes, u) and (\oplus, z) . Laws 13 and 14 relate these (otherwise independent) monoid structures, by saying that the unit of one is left absorbent of the other. Because of this left-bias, we might call these *left dioids* instead of just *dioids*.

Every bounded distributive lattice is a dioid for which (\otimes) commutes, perhaps the most classical example is Bool. The following is an instance in which (\otimes) does not commute:

```

type BinFun a = a × a -> a
instance Dioid (BinFun a) where
  z = λ(a, b) -> a
  u = λ(a, b) -> b
  f ⊕ g = λ(a, b) -> f (g (a, b), b)
  f ⊗ g = λ(a, b) -> f (a, g (a, b))

```

In models of deep-backtracking non-determinism, law 14 is replaced by a distribution.

$$(a \oplus b) \otimes c = (a \otimes c) \oplus (b \otimes c)$$

This makes the structure a near-semiring. See the work of Rivas, Jaskelioff and Schrijvers [16] for details. In this case it is possible to explore different results, whereas in the shallow case, we explore possible results in order only until one is found.

Let D_1 and D_2 be instances of Dioid. A *dioid homomorphism* is a function $f :: D_1 \rightarrow D_2$ such that the dioid instances are preserved:

$$\begin{aligned}
f (u ()) &= u () \\
f (z ()) &= z () \\
f (a \otimes b) &= f a \otimes f b \\
f (a \oplus b) &= f a \oplus f b
\end{aligned}$$

The type of f determines that the operations u , z , \otimes , and \oplus on the left-hand side come from the Dioid instance of

D_1 , while those on the right-hand side come from the Dioid instance of D_2 .

B. Categorification

Just as monoidal categories provide the right setting to express the notion of monoid in full generality, we now look for the analogous categorical structure to express the notion of dioid.

Definition III.1. A *dioid category* is a tuple $(\mathcal{C}, \otimes, I, \alpha_\otimes, \lambda_\otimes, \rho_\otimes, \oplus, Z, \alpha_\oplus, \lambda_\oplus, \rho_\oplus, \kappa_\otimes, \kappa_\oplus)$ where:

- \mathcal{C} is a category;
- $\otimes : \mathcal{C} \times \mathcal{C} \rightarrow \mathcal{C}$ and $\oplus : \mathcal{C} \times \mathcal{C} \rightarrow \mathcal{C}$ are bifunctors;
- I and Z are objects of \mathcal{C} ;
- $\alpha_\otimes, \lambda_\otimes, \rho_\otimes, \alpha_\oplus, \lambda_\oplus,$ and ρ_\oplus are natural isomorphisms:

$$\begin{aligned} \alpha_\otimes &: A \otimes (B \otimes C) \cong (A \otimes B) \otimes C \\ \lambda_\otimes &: I \otimes A \cong A \\ \rho_\otimes &: A \otimes I \cong A \\ \alpha_\oplus &: A \oplus (B \oplus C) \cong (A \oplus B) \oplus C \\ \lambda_\oplus &: Z \oplus A \cong A \\ \rho_\oplus &: A \oplus Z \cong A \end{aligned}$$

- κ_\otimes and κ_\oplus are natural transformations:

$$\begin{aligned} \kappa_\otimes &: Z \otimes A \rightarrow Z \\ \kappa_\oplus &: I \oplus A \rightarrow I \end{aligned}$$

Again, we expect these natural transformations to obey some coherence laws, which include those of $(\mathcal{C}, \otimes, I, \alpha_\otimes, \lambda_\otimes, \rho_\otimes)$ and $(\mathcal{C}, \oplus, Z, \alpha_\oplus, \lambda_\oplus, \rho_\oplus)$ being monoidal categories.

Given a dioid category, we can define what a dioid is.

Definition III.2. A *dioid in a dioid category* is an object D , together with operations $z : Z \rightarrow D$, $e : I \rightarrow D$, $s : D \oplus D \rightarrow D$ and $m : D \otimes D \rightarrow D$ for which the following laws hold:

$$m \circ (e \otimes \text{id}) = \lambda_\otimes \quad (15)$$

$$m \circ (\text{id} \otimes e) = \rho_\otimes \quad (16)$$

$$m \circ (\text{id} \otimes m) = m \circ (m \otimes \text{id}) \circ \alpha_\otimes \quad (17)$$

$$s \circ (z \oplus \text{id}) = \lambda_\oplus \quad (18)$$

$$s \circ (\text{id} \oplus z) = \rho_\oplus \quad (19)$$

$$s \circ (\text{id} \oplus s) = s \circ (s \oplus \text{id}) \circ \alpha_\oplus \quad (20)$$

$$m \circ (z \otimes \text{id}) = z \circ \kappa_\otimes \quad (21)$$

$$s \circ (e \oplus \text{id}) = e \circ \kappa_\oplus \quad (22)$$

The laws for dioids in dioid categories are the corresponding generalisations of equations for dioids.

To recover ordinary dioids, we take the category of sets and functions Set with $Z = I = \{*\}$ and $\oplus = \otimes = \times$. Notice that $\kappa_\otimes = \kappa_\oplus = \pi_1 : \{*\} \times A \rightarrow \{*\}$ is not a natural isomorphism, but only a natural transformation.

Lemma III.3. In general, from a monoidal category $(\mathcal{C}, \otimes, I, \alpha, \lambda, \rho)$ with Cartesian structure (terminal object and binary products), we obtain a dioid category

$(\mathcal{C}, \otimes, I, \alpha, \lambda, \rho, \times, 1, \alpha_\times, \pi_2, \pi_1, \kappa_\otimes, \pi_1)$, where:

$$\begin{aligned} \alpha_\times &= \langle \langle \pi_1, \pi_1 \circ \pi_2 \rangle, \pi_2 \circ \pi_2 \rangle : A \times (B \times C) \rightarrow (A \times B) \times C \\ \kappa_\otimes &= !_1 \otimes A : 1 \otimes A \rightarrow 1 \end{aligned}$$

We close our discussion on categorification of dioids by giving the generalisation of dioid homomorphism, which is a direct generalisation of that for ordinary dioids.

Definition III.4 (Dioid homomorphism). A *dioid homomorphism* from a dioid $(D_1, z_1, e_1, s_1, m_1)$ to a dioid $(D_2, z_2, e_2, s_2, m_2)$ is a morphism $f : D_1 \rightarrow D_2$ such that the following equations hold:

$$\begin{aligned} f \circ e_1 &= e_2 \\ f \circ z_1 &= z_2 \\ f \circ m_1 &= m_2 \circ (f \otimes f) \\ f \circ s_1 &= s_2 \circ (f \oplus f) \end{aligned}$$

C. Cartesian structure for functors

We can use Lemma III.3 to extend the monoidal categories of endofunctors discussed in Sections II-D and II-E to dioid categories. We need to establish that the category of endofunctors on a category \mathcal{C} has terminal object and binary products. If the base category \mathcal{C} has terminal object, then the constant functor to the terminal object is the terminal object on the category of endofunctors.

data K_1 **a where**

$$K_1 :: () \rightarrow K_1$$

instance Functor K_1 **where**

$$\text{fmap } f (K_1 ()) = K_1 ()$$

Here, the unit type $()$ represents the terminal object. Similarly, a product of endofunctors is defined in terms of product for objects in the base category, in a point-wise fashion:

data $(f \times g)$ **a where**

$$\text{Pair} :: f \times g \ a \rightarrow (f \times g) \ a$$

instance (Functor f , Functor g) \Rightarrow Functor $(f \times g)$ **where**

$$\text{fmap } h (\text{Pair } (fa, ga)) = \text{Pair } (\text{fmap } h \ fa, \text{fmap } h \ ga)$$

Thus, the endofunctors form a monoidal category with the Cartesian structure, and the monoidal categories supporting monads and applicative functors can be extended to dioid categories.

D. Non-determinism Monads

By Lemma III.3 and the Cartesian structure introduced above, we know that the category of endofunctors forms a dioid category by choosing tensor \otimes to be composition of functors \circ , tensor \oplus to be the binary product of functors \times , the object I to be the identity functor Identity and the object Z to be the constant terminal functor K_1 .

The dioids in this dioid category are functors d with operations ζ, η, σ and μ of type $K_1 \rightarrow d$, $\text{Identity} \rightarrow d$, $d \times d \rightarrow d$ and $d \circ d \rightarrow d$ respectively. If we expand the definitions, we can present this information in a type class:

class DioidM d **where**

$$\begin{aligned} \zeta &:: () \rightarrow d \ a \\ \eta &:: a \rightarrow d \ a \\ \sigma &:: d \ a \times d \ a \rightarrow d \ a \\ \mu &:: d \ (d \ a) \rightarrow d \ a \end{aligned}$$

subject to the laws:

$$\begin{aligned} \mu \circ \eta &= \text{id} \\ \mu \circ \text{fmap } \eta &= \text{id} \\ \mu \circ \text{fmap } \mu &= \mu \circ \mu \\ \sigma \circ \text{pair id zero} &= \text{fst} \\ \sigma \circ \text{pair zero id} &= \text{snd} \\ \sigma \circ \text{pair id } \sigma &= \sigma \circ \text{pair } \sigma \text{ id } \circ \alpha \\ \mu \circ \zeta &= \zeta \\ \sigma \circ \text{pair } \eta \text{ id} &= \eta \circ \text{fst} \end{aligned}$$

where $\text{pair } f \ g \ (x, y) = (f \ x, g \ y)$ and $\alpha \ (x, (y, z)) = ((x, y), z)$. The operations η and μ form an instance of Triple. In this way, a dioid might be seen as an extended monad. This justifies the equivalent type class

class Monad $m \Rightarrow$ MonadPlus m **where**

$$\begin{aligned} \text{mzero} &:: m \ a \\ \text{mplus} &:: m \ a \rightarrow m \ a \rightarrow m \ a \end{aligned}$$

subject to the following laws

$$\begin{aligned} \text{mplus mzero } u &= u \\ \text{mplus } u \ \text{mzero} &= u \\ \text{mplus } u \ (\text{mplus } v \ w) &= \text{mplus } (\text{mplus } u \ v) \ w \\ \text{mzero } \succ f &= \text{mzero} \\ \text{mplus } (\text{return } x) \ u &= \text{return } x \end{aligned}$$

in addition to those of monads.

Perhaps the most representative instance of MonadPlus subject to these axioms is Maybe.

instance MonadPlus Maybe **where**

$$\begin{aligned} \text{mzero} &= \text{Nothing} \\ \text{mplus Nothing } v &= v \\ \text{mplus (Just } x) \ v &= \text{Just } x \end{aligned}$$

An important non-example of MonadPlus subject to these axioms are lists. While the empty list and list concatenation would give an implementation for mzero and mplus, such implementation would not satisfy the law $\text{mplus } (\text{return } x) \ u = \text{return } x$. (In fact, lists are the canonical example of deep-backtracking non-determinism.)

E. Non-determinism Applicative Functors

We turn again to Lemma III.3 to obtain a dioid category of endofunctors, but this time with the Day convolution as a tensor instead of functor composition.

The dioids in this dioid category are functors d with operations empty , pure , $(\langle \rangle)$ and (\otimes) of type $K_1 \dot{\rightarrow} d$, $\text{Identity} \dot{\rightarrow} d$, $d \times d \dot{\rightarrow} d$ and $d * d \dot{\rightarrow} d$ respectively. If

we expand the definitions, we can present this information in a type class:

class DioidF d **where**

$$\begin{aligned} \text{empty} &:: () \rightarrow d \ a \\ \text{pure} &:: a \rightarrow d \ a \\ (\langle \rangle) &:: d \ a \times d \ a \rightarrow d \ a \\ (\otimes) &:: d \ (b \rightarrow a) \times d \ b \rightarrow d \ a \end{aligned}$$

As we did with monads, we separate the applicative functor contained in this type class, and create a class that extends applicative functors with the additional information:

class Applicative $f \Rightarrow$ Alternative f **where**

$$\begin{aligned} \text{empty} &:: f \ a \\ (\langle \rangle) &:: f \ a \rightarrow f \ a \rightarrow f \ a \end{aligned}$$

By instantiating the laws for dioids, we obtain the following laws for Alternative, which are additional to those of the underlying Applicative instance.

$$\begin{aligned} \text{empty } \langle \rangle \ u &= u \\ u \ \langle \rangle \ \text{empty} &= u \\ u \ \langle \rangle \ (v \ \langle \rangle \ w) &= (u \ \langle \rangle \ v) \ \langle \rangle \ w \\ \text{empty } \otimes \ u &= \text{empty} \\ \text{pure } x \ \langle \rangle \ u &= \text{pure } x \end{aligned}$$

We can extend Remark II.4 to the type classes MonadPlus and Alternative, and obtain the following result.

instance MonadPlus $m \Rightarrow$ Alternative m **where**

$$\begin{aligned} \text{empty} &= \text{mzero} \\ u \ \langle \rangle \ v &= \text{mplus } u \ v \end{aligned}$$

In fact, most Alternative instances found in programming practice are actually MonadPlus instances. An example of an Alternative which is not a MonadPlus is the constant functor on a Dioid.

instance Dioid $d \Rightarrow$ Alternative $(K \ d)$ **where**

$$\begin{aligned} \text{empty} &= \text{MK } z \\ (\text{MK } d_1) \ \langle \rangle \ (\text{MK } d_2) &= \text{MK } (d_1 \oplus d_2) \end{aligned}$$

IV. FREE STRUCTURES

Free structures are a fundamental tool in universal algebra, as in some sense they provide the most *general* models of an algebraic structure, free of any additional equation over terms. In computer science, this phenomenon is often referred to as the *no junk, no confusion* principle [20]. In our setting, we employ free structures as a device to work with those programs that only use the structure under analysis.

A. Free monoids

The notion of free ordinary monoid is captured by a universal property. Formally, we say that the type FreeMon_a is the free monoid over a when:

- FreeMon_a is an instance of Monoid;
- there is a function $\text{ins} :: a \rightarrow \text{FreeMon}_a$;

- for any Monoid instance m and function $f :: a \rightarrow m$, there exists a unique monoid homomorphism $\text{univ } f :: \text{FreeMon}_a \rightarrow m$ such that $\text{univ } f \circ \text{ins} = f$.

While mathematically precise, this definition is not constructive: it does not provide a procedure to construct such FreeMon_a from a given a . A possible technique to find a concrete construction is to provide a unique form for monoidal terms, such that two terms that are equal by the monoid laws are represented by the same term. For example, $a \otimes (u () \otimes (b \otimes c))$ and $(a \otimes b) \otimes (c \otimes u ())$ should have a unique representation in the data type representing the free monoid over a set which includes a , b and c . To see that two monoid expressions are equivalent, we can apply the monoid laws in a methodological way:

- every atom a is replaced by $a \otimes u ()$;
- every expression associated to the left is re-associated to the right;
- every expression $u () \otimes t$ is reduced to t .

Applying this method to the expressions above, we obtain the term $a \otimes (b \otimes (c \otimes u ()))$ in both cases. After some thinking, one can conclude that every term reduces to a list of atoms ending in $u ()$. This observation inspires the following data type for representing canonical forms.

data FreeMon a where

```
Nil :: FreeMon a
Cons :: a × FreeMon a → FreeMon a
```

This data type is equivalent to a list of as , and therefore has a monoid instance given by the empty list and list concatenation:

instance Monoid (FreeMon a) where

```
u () = Nil
Nil ⊗ bs = bs
(Cons a as) ⊗ bs = Cons a (as ⊗ bs)
```

The insertion function represents an a atom by a singleton list.

```
ins :: a → FreeMon a
ins a = Cons a Nil
```

The function univ is written by recursion on $\text{FreeMon } a$:

```
univ :: Monoid m ⇒ (a → m) → FreeMon a → m
univ f Nil = u ()
univ f (Cons a as) = f a ⊗ univ f as
```

Using set theory, the free monoid over a set a , i.e. lists of a , can be seen as the least solution a^* to the recursive equation:

$$a^* = \{*\} \cup a \times a^*$$

where \cup represents the disjoint union of sets.


Generalising to monoidal categories the equation becomes

$$A^* \cong I + A \otimes A^* \quad (23)$$

which gives a candidate for the free monoid in a monoidal category. Before instantiating this formula to other cases, we

first review the general universal property for a free monoid in a monoidal category $(\mathcal{C}, \otimes, I)$.

Definition IV.1 (Free monoid). Let X be an object, the *free monoid* over X is a monoid (F, e_F, m_F) together with a morphism $\text{ins} : X \rightarrow F$ such that for any monoid (M, e_M, m_M) and morphism $f : X \rightarrow M$ there exists a unique monoid homomorphism $\text{univ}(f) : F \rightarrow M$ such that $\text{univ}(f) \circ \text{ins} = f$. Diagrammatically, we have:



The morphism ins is called the *insertion* morphism, and $\text{univ } f$ is known as the *lifting* of f .

As in the case of ordinary monoids, this definition provides an abstract characterisation for the free monoid. To obtain a concrete description, we instantiate formula 23 to the corresponding monoidal category.

For obtaining the free Monad, we apply formula 23 to the monoidal category of endofunctors with functor composition as tensor, which yields the equation

$$f^* \cong \text{Identity} + f \circ f^*$$

that leads to the following data type:

data Free f a where

```
Nilo :: a → Free f a
Conso :: f (Free f a) → Free f a
```

This is indeed the free monad on a functor f , with monad instance:

instance Functor $f \Rightarrow$ Monad (Free f) where

```
return x = Nilo x
Nilo x ≻ f = f x
Conso v ≻ f = Conso (fmap (≻ f) v)
```

The insertion morphism and lifting are as follows:

```
ins :: Functor f ⇒ f → Free f
ins v = Conso (fmap Nilo v)
univ :: (Functor f, Monad m) ⇒ (f → m) → (Free f → m)
univ f (Nilo x) = return x
univ f (Conso v) = f (fmap (univ f) v) ≻ id
```

We now turn our focus to the monoidal category of endofunctors with Day convolution as tensor. Instantiating formula 23 to this monoidal category results in:

$$f^* \cong \text{Identity} + f * f^*$$

which leads to the following data type

data FreeA f a where

```
Nil* :: a → FreeA f a
Cons* :: f (b → a) × FreeA f b → FreeA f a
```


The insertion morphism and lifting are as follows:

```

ins :: a → Free a
ins a = Term (LiftT a)
univ :: Dioid d ⇒ (a → d) → Free a → d
univ f Zero     = z ()
univ f Unit     = u ()
univ f (Term v) = univT f v
univT :: Dioid d ⇒ (a → d) → Term a → d
univT f (LiftT x)      = f x
univT f (SumT1 s)    = univM f s ⊕ u ()
univT f (SumT2 (s, t)) = univM f s ⊕ univT f t
univT f (MultT1 s)    = univS f s ⊗ z ()
univT f (MultT2 (s, t)) = univS f s ⊗ univT f t

```

where univ_M and univ_S are auxiliary functions that work as expected.

The definition of the free dioid and its operations is not complicated but is rather subtle. Therefore we have formally verified that this is indeed the free dioid using Agda as a proof assistant.

We now turn to the problem of generalising the equations (24–27) to other dioid categories. For generalising the free monoid construction, we replaced disjoint union \sqcup for coproduct $+$, Cartesian product \times for monoidal tensor \otimes , and the singleton set $\{*\}$ for the unit object I of a monoidal category. In the case of dioids, the system of recursive equations presented only involve, in addition to disjoint union, Cartesian product \times and the singleton set $\{*\}$. Nevertheless, a dioid category has more structure: two objects Z and I , and two bifunctors \oplus and \otimes . That is the reason we introduced the renamings \times_\otimes , \times_\oplus , 0 , and 1 : for keeping track of which constructors correspond to the additive structure and which to the multiplicative structure.

Replacing \sqcup for $+$, \times_\otimes for \otimes , 1 for I , \times_\oplus for \oplus , and 0 for Z in the equations, we obtain the following system of equations for the tentative free dioid over an object X :

$$\begin{aligned}
X^* &= Z + I + T \\
T &= X + (S \oplus I) + (S \oplus T) + (M \otimes Z) + (M \otimes T) \\
S &= X + (M \otimes Z) + (M \otimes T) \\
M &= X + (S \oplus I) + (S \oplus T)
\end{aligned}$$

When considering the category of endofunctors with composition as multiplication and Cartesian product as addition, we obtain data type constructors representing these formulas:

```

data Freeo f x where
  Zeroo :: K1 x → Freeo f x
  Unito  :: Identity x → Freeo f x
  Termo :: Termo f x → Freeo f x
data Termo f x where
  LiftTo :: f x → Termo f x
  SumTo1 :: Sumo f x × Identity x → Termo f x
  SumTo2 :: Sumo f x × Termo f x → Termo f x
  MultTo1 :: Multo f (K1 x) → Termo f x

```

$$\text{MultT}_o^2 :: \text{Mult}_o f (\text{Term}_o f x) \rightarrow \text{Term}_o f x$$

```

data Sumo f x where
  LiftSo  :: f x → Sumo f x
  MultSo1 :: Multo f (K1 x) → Sumo f x
  MultSo2 :: Multo f (Termo f x) → Sumo f x
data Multo f x where
  LiftMo  :: f x → Multo f x
  SumMo1  :: Sumo f x × Identity x → Multo f x
  SumMo2  :: Sumo f x × Termo f x → Multo f x

```

While at first sight this data type seemed to work, we ran into problems when we tried to define the MonadPlus instance. Specifically, we got stuck when writing the $(\>=)$ operator on $\text{Free}_o f$, as we needed to distribute coproducts on the right of \circ . Our conjecture is that the recursive equations presented only work for tensors which distribute coproducts on both parameters. Fortunately, Day convolution satisfies such requirement, and we successfully applied the formulas to obtain the free Alternative . The resulting data type constructor is similar to that of monads, but we inlined the data type constructors K_1 and Identity to avoid some clutter:

```

data Free* f x where
  Zero*  :: Free* f x
  Unit*  :: x → Free* f x
  Term*  :: Term* f x → Free* f x
data Term* f x where
  LiftT*  :: f x → Term* f x
  SumT*1  :: Sum* f x → x → Term* f x
  SumT*2  :: Sum* f x → Term* f x → Term* f x
  MultT*1 :: Mult* f (a → x) → Term* f x
  MultT*2 :: Mult* f (a → x) → Term* f a → Term* f x
data Sum* f x where
  LiftS*  :: f x → Sum* f x
  MultS*1 :: Mult* f (a → x) → Sum* f x
  MultS*2 :: Mult* f (a → x) → Term* f a → Sum* f x
data Mult* f x where
  LiftM*  :: f x → Mult* f x
  SumM*1  :: Sum* f x → x → Mult* f x
  SumM*2  :: Sum* f x → Term* f x → Mult* f x

```

The instances of Functor , Applicative and Alternative are involved, although they follow the same pattern of the operations for free ordinary dioids. We present the Applicative instance, as its multiplication is the most involved operation.

```

instance Functor f ⇒ Applicative (Free* f) where
  pure x = Unit* x
  Zero* ⊗ v = Zero*
  Unit* f ⊗ x = fmap f x
  Term* x ⊗ y = Term* (x ⊗T y)

```

The function (\otimes_T) is the equivalent to (\otimes_T) for free ordinary dioids, and it is defined as follows:

```

(⊗T) :: Functor f ⇒
  Term* f (a → b) → Free* f a → Term* f b

```

$$\begin{aligned}
\text{LiftT}_* x \otimes_T \text{Zero}_* &= \text{MultT}_*^1 (\text{LiftM}_* x) \\
\text{LiftT}_* x \otimes_T \text{Unit}_* y &= \text{LiftT}_* (\text{eval}_F x y) \\
\text{LiftT}_* x \otimes_T \text{Term}_* y &= \text{MultT}_*^2 (\text{LiftM}_* x) y \\
\text{SumT}_*^1 x y \otimes_T \text{Zero}_* &= \text{MultT}_*^1 (\text{SumM}_*^1 x y) \\
\text{SumT}_*^1 x y \otimes_T \text{Unit}_* v &= \text{SumT}_*^1 (\text{eval}_F x v) (y v) \\
\text{SumT}_*^1 x y \otimes_T \text{Term}_* v &= \text{MultT}_*^2 (\text{SumM}_*^1 x y) v \\
\text{SumT}_*^2 x v \otimes_T \text{Zero}_* &= \text{MultT}_*^1 (\text{SumM}_*^2 x v) \\
\text{SumT}_*^2 x v \otimes_T \text{Unit}_* y &= \text{SumT}_*^2 (\text{eval}_F x y) (\text{eval}_F v y) \\
\text{SumT}_*^2 x v \otimes_T \text{Term}_* y &= \text{MultT}_*^2 (\text{SumM}_*^2 x v) y \\
\text{MultT}_*^1 x \otimes_T v &= \text{MultT}_*^1 (\text{fmap uncurry } x) \\
\text{MultT}_*^2 x v \otimes_T w &= \text{MultT}_*^2 (\text{fmap uncurry } x) \\
&\quad (\text{fmap } (,) v \otimes_T w)
\end{aligned}$$

$$\begin{aligned}
\text{eval}_F &:: \text{Functor } f \Rightarrow f (a \rightarrow b) \rightarrow a \rightarrow f b \\
\text{eval}_F v a &= \text{fmap } (\lambda f \rightarrow f a) v
\end{aligned}$$

The lifting function is also involved, although we insist again that it follows the pattern presented for ordinary dioids. We close our discussion for free structures by presenting the insertion function for Free_* .

$$\begin{aligned}
\text{ins}_* &:: \text{Functor } f \Rightarrow f a \rightarrow \text{Free}_* f a \\
\text{ins}_* v &= \text{Term}_* (\text{LiftT}_* v)
\end{aligned}$$

The correctness for these definitions was also formally verified using Agda.

V. CONCLUSION

This paper has introduced a generalised notion of dioids, which was used to study computations with shallow-backtracking non-determinism. By considering MonadPlus and Alternative type classes as dioids in the category of endofunctors, we have obtained a set of laws that those type classes should obey.

We have shown a concrete description for the free ordinary dioid, and then generalised the construction to dioid categories. By instantiation, this resulted in the construction of the free Alternative, but we were not able to do the same to obtain a free MonadPlus .

As further work, it would be interesting to study the notion of dioid in the dioid category of endoprofunctors, and see whether the general formulation for free dioids presented can be used in this setting to obtain a free arrow with a notion of shallow-backtracking non-determinism.

ACKNOWLEDGMENT

The authors are grateful to the members of IFIP WG 2.1 for preliminary discussions about the topic of this paper.

REFERENCES

- [1] E. Moggi, "Computational lambda-calculus and monads," in *Proceedings of the Fourth Annual Symposium on Logic in Computer Science*. IEEE Press, 1989, pp. 14–23. [Online]. Available: <http://dl.acm.org/citation.cfm?id=77350.77353>
- [2] —, "Notions of computation and monads," *Information and Computation*, vol. 93, no. 1, pp. 55–92, Jul. 1991. [Online]. Available: [http://dx.doi.org/10.1016/0890-5401\(91\)90052-4](http://dx.doi.org/10.1016/0890-5401(91)90052-4)
- [3] P. Wadler, "Comprehending monads," in *Proceedings of the 1990 ACM Conference on LISP and Functional Programming*, ser. LFP '90. New York, NY, USA: ACM, 1990, pp. 61–78. [Online]. Available: <http://doi.acm.org/10.1145/91556.91592>
- [4] M. Spivey, "A functional theory of exceptions," *Sci. Comput. Program.*, vol. 14, no. 1, pp. 25–42, May 1990. [Online]. Available: [https://doi.org/10.1016/0167-6423\(90\)90056-J](https://doi.org/10.1016/0167-6423(90)90056-J)
- [5] C. McBride and R. Paterson, "Applicative programming with effects," *Journal of Functional Programming*, vol. 18, no. 01, pp. 1–13, 2008. [Online]. Available: <http://dx.doi.org/10.1017/S0956796807006326>
- [6] S. D. Swierstra and L. Duponcheel, "Deterministic, error-correcting combinator parsers," in *Advanced Functional Programming, Second International School, Olympia, WA, USA, August 26-30, 1996, Tutorial Text*, ser. Lecture Notes in Computer Science, J. Launchbury, E. Meijer, and T. Sheard, Eds., vol. 1129. Springer, 1996, pp. 184–207. [Online]. Available: http://dx.doi.org/10.1007/3-540-61628-4_7
- [7] J. Gibbons and B. c. d. s. Oliveira, "The essence of the iterator pattern," *J. Funct. Program.*, vol. 19, no. 3-4, pp. 377–402, Jul. 2009. [Online]. Available: <http://dx.doi.org/10.1017/S0956796809007291>
- [8] M. Jaskelioff and O. Rypacek, "An investigation of the laws of traversals," in *Proceedings Fourth Workshop on Mathematically Structured Functional Programming*, Tallinn, Estonia, 25 March 2012, ser. Electronic Proceedings in Theoretical Computer Science, J. Chapman and P. B. Levy, Eds., vol. 76. Open Publishing Association, 2012, pp. 40–49. [Online]. Available: <https://arxiv.org/abs/1202.2919v1>
- [9] S. Marlow, L. Brandy, J. Coens, and J. Purdy, "There is no fork: An abstraction for efficient, concurrent, and concise data access," in *Proceedings of the 19th ACM SIGPLAN International Conference on Functional Programming*, ser. ICFP '14. New York, NY, USA: ACM, 2014, pp. 325–337. [Online]. Available: <http://doi.acm.org/10.1145/2628136.2628144>
- [10] E. Rivas and M. Jaskelioff, "Notions of computation as monoids," *CoRR*, vol. abs/1406.4823, 2014. [Online]. Available: <http://arxiv.org/abs/1406.4823>
- [11] S. Mac Lane, *Categories for the Working Mathematician*, ser. Graduate Texts in Mathematics. Springer-Verlag, 1971, no. 5, second edition, 1998.
- [12] J. Hughes, "A novel representation of lists and its application to the function "reverse";," *Information Processing Letters*, vol. 22, no. 3, pp. 141–144, 1986. [Online]. Available: [http://dx.doi.org/10.1016/0020-0190\(86\)90059-1](http://dx.doi.org/10.1016/0020-0190(86)90059-1)
- [13] J. Voigtländer, "Asymptotic improvement of computations over free monads," in *Proceedings of the 9th International Conference on Mathematics of Program Construction*, ser. MPC '08. Springer-Verlag, 2008, pp. 388–403. [Online]. Available: http://dx.doi.org/10.1007/978-3-540-70594-9_20
- [14] T. Uustalu, "A divertimento on monadplus and nondeterminism," *J. Log. Algebr. Meth. Program.*, vol. 85, no. 5, pp. 1086–1094, 2016. [Online]. Available: <http://dx.doi.org/10.1016/j.jlamp.2016.06.004>
- [15] J. M. Spivey, "When maybe is not good enough," *J. Funct. Program.*, vol. 22, no. 6, pp. 747–756, 2012. [Online]. Available: <http://dblp.uni-trier.de/db/journals/jfp/jfp22.html#Spivey12>
- [16] E. Rivas, M. Jaskelioff, and T. Schrijvers, "From monoids to nearsemirings: The essence of monadplus and alternative," in *Proceedings of the 17th International Symposium on Principles and Practice of Declarative Programming*, ser. PPDP'15, M. Falaschi and E. Albert, Eds. ACM, 2015, pp. 196–207. [Online]. Available: <http://doi.acm.org/10.1145/2790449.2790514>
- [17] M. Grandis, "Cubical monads and their symmetries," *Rend. Instit. Mat. Univ. Trieste*, vol. 25, pp. 223–264, 1993. [Online]. Available: <http://hdl.handle.net/10077/4693>
- [18] E. Bainbridge, P. J. Freyd, A. Scedrov, and P. J. Scott, "Functorial polymorphism," *Theoretical Computer Science*, vol. 70, no. 1, pp. 35–64, 1990. [Online]. Available: <http://www.sciencedirect.com/science/article/pii/0304397590901517>
- [19] B. Day, "Note on monoidal localisation," *Bulletin of the Australian Mathematical Society*, vol. 8, pp. 1–16, 2 1973. [Online]. Available: http://journals.cambridge.org/article_S0004972700045433
- [20] R. M. Burstall and J. A. Goguen, *Algebras, Theories and Freeness: An Introduction for Computer Scientists*. Dordrecht: Springer Netherlands, 1982, pp. 329–349. [Online]. Available: http://dx.doi.org/10.1007/978-94-009-7893-5_11
- [21] G. M. Kelly, "A unified treatment of transfinite constructions for free algebras, free monoids, colimits, associated sheaves, and so on," *Bulletin of the Australian Mathematical Society*, vol. 22, no. 01, pp. 1–83, 1980. [Online]. Available: <https://doi.org/10.1017/S0004972700006353>